CheckBox REF Substitute
by Charles Kenyon
Normally, you can use a REF field to duplicate the contents of a formfield elsewhere in a document. Unfortunately this doesn’t work with checkbox fields. The following on-exit macro will work with the nine checkboxes named Check1, Check2, Check3, Check1a, Check2a, Check3a, Check1b, Check2b, and Check3b. The first three have the macro listed as their on-exit macro; the checkboxes in the second and third groups of three are not "enabled." That means that although they are checkbox formfields, they can't be checked or unchecked independently by the user; the only way they are changed is with the VBA code or by setting their "checked / unchecked" property while the document is unprotected.

The macro listing is:

===========

Sub CopyCheckBoxMacro()

'   Macro written by Charles Kenyon

'   21 December 2001 modified 11 January 2002

'

'   Declare variables

    Dim sCheck As String

    Dim bCheck As Boolean

    Dim sChecka As String

    Dim sCheckb As String

'

'   Get name and value of formfield which was just used and

'       set variable for dependent formfield name

    sCheck = Selection.FormFields.Item(1).Name

    sChecka = sCheck & "a"

    sCheckb = sCheck & "b"

    bCheck = ActiveDocument.FormFields(sCheck).CheckBox.Value

'

'   Unprotect document

    ActiveDocument.Unprotect

'

'   Set value for dependent checkboxes and reprotect document

    ActiveDocument.FormFields(sChecka).CheckBox.Value = bCheck

    ActiveDocument.FormFields(sCheckb).CheckBox.Value = bCheck

    ActiveDocument.Protect Type:=wdAllowOnlyFormFields, NoReset:=True

End Sub

============

This certainly is not as handy as using the REF field, but you should be able to adapt it to your needs. It is designed so that it can be used with multiple checkboxes rather than having a different macro for each checkbox. The drawback is that it expects that each checkbox will have the same number (in this case 1) of dependent checkboxes.

This document is protected for filling in forms (locked) without a password. The macros depend on that state.

This tutorial can be downloaded as a part of the Checkbox template from my download page. This is version 2023.11.

